

Estimation of relative potency in a 4-point, parallel line assay

Stefano Vezzoli, CROS NT, Verona, Italy

ABSTRACT

For inhaled asthma medications, the current EMA guideline requires equivalence in respect of efficacy to be demonstrated also through a comparison between drugs in terms of relative potency. Differently from the more common comparison of the magnitude of responses observed following administration of different products, the dose-response relationship is taken into account by this approach. In this paper, we discuss the rationale of a two-step analysis to estimate relative potency and its confidence limits in a 4-point assay, when the same two dose levels of each product being compared are administered. This is the typical design for the comparison of different formulations of the same medication. The analysis is based on the assumption of a parallel line assay, i.e. linear and parallel dose-response curves on log-dose scale. The development of a SAS® macro to handle this analysis in both parallel group and cross-over designs will also be described.

INTRODUCTION

In the current EMA Guideline on the Requirements for Clinical Documentation for Orally Inhaled Products [1], two different approaches for the demonstration of therapeutic equivalence in respect of efficacy of inhaled asthma drugs are described: demonstration of equivalence for at least two dose levels on the pharmacodynamic endpoint and analysis of relative potency. The relative potency of the test product to the reference product is defined as the dose of the test product that produces the same biological response as one unit of the dose of the reference product. The first approach is based on the usual comparison on the response axis, while the second one focuses on the dose axis, allowing the dose-response relationship to be taken into account. For either approach to be acceptable a minimum requirement is that the study has assay sensitivity, therefore at least two non-zero dose levels of both the test and reference products need to be studied and one dose level needs to be shown to be superior to the other. In this paper, the analysis of relative potency will be illustrated in a 4-point assay, when the same two dose levels of each product being compared are administered. This is the typical design for the comparison of different formulations of the same medication, for example a dry powder inhaler and a pressurized metered-dose inhaler. The analysis is based on the assumption of a parallel line assay, i.e. linear and parallel dose-response curves on log-dose scale. A continuous response is also assumed. A two-step approach to estimate relative potency and its confidence limits in both parallel group and cross-over designs will be discussed and a SAS macro to implement this method will be presented.

ESTIMATION OF RELATIVE POTENCY

All the following considerations regarding the point and interval estimation of relative potency apply to all the parallel line assays and are not limited to the 4-point design.


POINT ESTIMATE

If we define X_T and X_R as the doses of the test and the reference products producing the same response, then the relative potency is given by $\rho = X_T/X_R$. Based on this definition, if relative potency is:

- < 1 , then a dose of the test produces the same result as does a higher dose of the reference (test is more potent);
- $= 1$, then the two products produce the same result at the same dose (test and reference are equipotent);
- > 1 , then a dose of the test produces the same result as does a lower dose of the reference (test is less potent).

In Figure 1, an example of dose-response curves for two products based on the E_{\max} model [2] is provided. At the response level highlighted in the figure, $X_T > X_R$, therefore the relative potency is > 1 and the test is less potent than the product. However, it is clear from Figure 1 that these dose-response curves lead to generally different values for relative potency depending on the response level considered.


Figure 1. Dose-response curves of test (grey) and reference (black) products on the original dose scale. The doses X_T of the test and X_R of the reference give the same response.



In order to postulate a unique value for relative potency and to allow its estimation, we assume a parallel line assay, i.e. linear and parallel dose-response curves on log-dose scale. Alternative assumptions can be made (e.g., slope-ratio assay: linear dose-response curves on the original scale, with the same intercept and different slopes), but they are not considered in the present paper.

As an example, a portion of the same dose-response curves of Figure 1 is represented in Figure 2 on the log-dose scale. We observe that in the region highlighted in Figure 2 the assumption of parallel line assay seems to hold. This result was expected, since under the E_{max} model the log-transformation of the dose linearises the dose-response relationship when the effect ranges from 20% to 80% of the maximum response [3]. It should be noted that the example provided is based on the E_{max} model just for its widespread use, but the method presented in this paper can be applied also with different shapes of the dose-response, as long as the required assumptions are satisfied.

Figure 2. Dose-response curves on the log-dose scale. The dashed square indicates the region where the assumption of parallel line assay holds.



On the log-dose scale, the horizontal distance between the linear and parallel dose-response curves is constant. It follows that, if we define $x_T = \log(X_T)$ and $x_R = \log(X_R)$ (in the present paper \log denotes the natural logarithm, but the choice of the base of the logarithm is irrelevant as long as it is consistent with the back-transformation applied at the end of the calculations), the difference $x_T - x_R$ (corresponding to the horizontal distance) and therefore the ratio X_T/X_R are constant. This condition ensures a unique value for relative potency, irrespective of the response level considered.

In order to express the relative potency in terms of estimable quantities, we now focus on the difference $x_T - x_R$. This difference corresponds to the logarithm of the relative potency ρ , since

$$\log(\rho) = \log(X_T/X_R) = \log(X_T) - \log(X_R) = x_T - x_R.$$

The regression lines for the dose-response curves of the treatment and reference products can be expressed respectively as


$$Y_T = \alpha_T + \beta \cdot x_T \quad \text{and} \quad Y_R = \alpha_R + \beta \cdot x_R,$$

where Y_i is the response, α_i is the intercept ($i = T, R$) and β is the common slope. The log relative potency is the difference $x_T - x_R$ when $Y_T = Y_R$. By equating the responses of the two products, we obtain

$$\begin{aligned} Y_T &= Y_R \\ \alpha_T + \beta \cdot x_T &= \alpha_R + \beta \cdot x_R \\ \log(p) &= x_T - x_R = (\alpha_R - \alpha_T) / \beta. \end{aligned}$$

The terms included in the above expressions are graphically represented in Figure 3. The log relative potency can be thus estimated as the ratio between the estimate of the overall product effect $\alpha_R - \alpha_T$ and the estimate of the common slope of $\log(\text{dose})$ β : $(\alpha_R - \alpha_T) / \beta$. The same formula for the estimation of relative potency is provided in the Health Canada guidance to establish equivalence of Short-Acting β_2 -Agonists (SABA) in asthma [4].

Figure 3. Dose-response curves of test (grey) and reference (black) products on the log-dose scale. Graphical representation of the terms involved in the derivation of $\log(p)$.



CONFIDENCE INTERVAL

The calculation of the confidence interval (CI) of log relative potency is based on Fieller's theorem [4,5]. This covers a general situation in which the CI of a ratio between two variables following either exactly or approximately a bivariate normal distribution with a non-zero covariance is of interest. The confidence limits for $\log(p)$ are then

$$\frac{R - \frac{g v_{12}}{v_{22}} \pm \frac{t}{b} \left[v_{11} - 2R v_{12} + R^2 v_{22} - g \left(v_{11} - \frac{v_{12}^2}{v_{22}} \right) \right]^{1/2}}{(1 - g)},$$

where:

- R is the estimate $(\alpha_R - \alpha_T) / b$ of log relative potency;
- $g = t^2 v_{22} / b^2$;
- v_{11} is the variance of $\alpha_R - \alpha_T$;
- v_{12} is the covariance between $\alpha_R - \alpha_T$ and b ;
- v_{22} is the variance of b
- t is the appropriate percentile of the t -distribution with f degrees of freedom;
- f are the degrees of freedom on which the residual variance is based.

Estimates of relative potency and its CI are finally obtained by exponentiating R and the confidence limits calculated using Fieller's theorem.

The EMA guideline requires the CI of relative potency to lie entirely within the limits 0.67 – 1.5 [1]. Unfortunately, the confidence level to be used for the interval is not explicitly specified in the guideline. A clarification on this point has been recently provided by García-Arieta, an assessor of the Spanish Agency for Medicines and Health Products

(Agencia Española de Medicamentos y Productos Sanitarios, AEMPS): “*relative potency confidence interval can be calculated at a 90% confidence level like pharmacokinetic bioequivalence studies*” [6].

ASSUMPTIONS FOR THE ESTIMATION OF RELATIVE POTENCY

The estimation of relative potency in a parallel line assay requires the following main assumptions to be satisfied [7]:

- A. absence of significant deviations from parallelism of dose-response curves on the log-dose scale. In the previous section it has already been clarified that in case of non-parallel curves a unique value for relative potency cannot be assumed. Estimates of relative potency obtained when this condition does not hold must be discarded;
- B. significant dose-response relationship. If the slope b is not significantly different from zero, the data are consistent with a zero value for β and hence an infinite value for the relative potency. Furthermore, the EMA guideline explicitly requires the assay sensitivity of the study [1];
- C. absence of a significant difference between products. The difference $a_R - a_T$ should be non-significant. If this assumption does not hold, but relative potency is estimated anyway, the estimate obtained is expected to be far from unity.

Linearity is another crucial assumption in parallel line assays, but obviously it cannot be tested in a 4-point design.

An entire section of the classical textbook by Finney is devoted to the choice of the significance level to be considered when testing the above assumptions [7] and the reader interested in the issue is encouraged to consult this reference. In our limited experience, we considered the usual significance level of 0.05 for assumptions B and C and a more stringent level of 0.1 for the fundamental assumption A.

THE TWO-STEP ANALYSIS

The proposed method for the analysis of relative potency is based on a two-step approach. From now on, we will strictly focus on 4-point parallel line assays, when the same two dose levels of each product being compared are administered.

STEP 1

In the first step a model is estimated in order to verify the required assumptions. The treatment effect is decomposed into the effects of product, dose on the log-scale and their interaction. With this aim, the 4-level categorical variable commonly used for treatment is replaced by the following variables: product (two levels, e.g. product A vs. product B) and logarithm of the dose, to be included in the model with their interaction. An example is provided in the following table.

Former variable	New variables	
Treatment	Product	log(dose)
Product A - 50 µg	A	3.912
Product A - 200 µg	A	5.298
Product B - 50 µg	B	3.912
Product B - 200 µg	B	5.298

If we considered combination products with two active substances, we would have several choices for the definition of log(dose). For example, if the doses tested were 50/10 µg (1 puff of the combination) vs. 200/40 µg (4 puffs), possible values for log(dose) would be 3.912 vs. 5.298 (dose of the first substance of the combination) or 2.303 vs. 3.689 (dose of the second substance) or 0 vs. 1.386 (number of puffs). The choice is completely irrelevant, since the ratio of the values submitted to log transformation is always 4.

All the other appropriate factors (e.g., baseline, centre, country, subject, period, etc.) can be kept in the model as usual. The approach presented assumes that no interaction terms involving the treatment (e.g., treatment * centre, treatment * visit) are to be included in the model. Therefore the proposed method can be applied to a wide range of models in both parallel group and cross-over designs.

Based on the estimated model, the assumptions are checked (the significance levels used should not be considered as mandatory, see also the previous section):

- A. absence of significant deviations from parallelism: the p-value of the interaction product * log(dose) should be ≥ 0.1 ;
- B. significant dose-response relationship: the p-value of the slope for log(dose) should be < 0.05 ;
- C. absence of a significant difference between products: the p-value of the product effect should be ≥ 0.05 .

Regarding assumption C, it should be noted that, since the model includes the interaction term, the estimate of the product effect and its significance depend on the dose level considered. The product effect corresponds to the vertical distance between the dose-response curves on the log-dose scale and, since non-parallelism is allowed by the model, this distance is not generally constant. For this reason we suggest to test the product effect at the two log(dose) levels corresponding to the doses actually administered in the study and at the mean log(dose) level. In absence of significant deviations from parallelism, these three tests are expected to lead to consistent results.

In absence of evidence against the assumptions, we can proceed to the second step.

STEP 2

The same model used in step 1 is estimated excluding the non-significant term for interaction. Assumptions B and C may be checked for confirmation. The absence of a significant difference between products can now be verified without taking into account particular dose levels, since the vertical distance between the dose-response curves is constant due to the exclusion of the interaction term from the model. Then log relative potency can be estimated as the ratio between the estimated product effect (of note, the difference R-T should be considered instead of the usual T-R) and the estimated slope for log(dose): $(a_R - a_T) / b$. Its confidence limits can be obtained based on the other results from the model.

THE %REL POT MACRO

The %REL POT macro is designed to implement the two-step approach described for the calculation of relative potency and its confidence interval. The macro call allows the user to set up quickly the necessary parameters for the calculation.

```
%relpot(data=, response=, fixed_cov=, random_cov=, categorical_cov=, alpha=);
```

Macro %REL POT contains six parameters:

- data: name of the dataset to be analysed. The variables “product” and “dose” are required. The product has to be coded using the character strings “T” for the test and “R” for the reference. The numeric variable “dose” is the original (i.e., untransformed) dose. For example:

	subject	country	treatment	product	dose	baseline	y
1	1	3	T 50	T	50	3.71	6.47
2	149	2	T 200	T	200	6.96	10.01
3	229	2	R 50	R	50	5.8	6.95
4	315	3	R 200	R	200	7.37	9.24

- response: name of the response variable in the dataset;
- fixed_cov: names of the variables to be included as fixed effects in the model. The variables “product” and “dose” will be automatically included, therefore they should not be specified here;
- random_cov: names of the variables to be included as random effects in the model;
- categorical_cov: names of the categorical variables included in the model. Both fixed and random effects should be considered. The variable “product” will be automatically included, therefore it should not be specified here;
- alpha: $1-\alpha$ is the confidence level of the interval. For example, if the 90% CI is required, alpha should be set at 0.1.

The outline of the macro is as follows:

1. All the datasets in the Work library with the name prefix “rp” are deleted, since this prefix is used by the datasets generated by the macro.
2. The log transformed doses are derived. The two log(dose) levels corresponding to the doses actually administered in the study and the mean log(dose) level are calculated and stored as macro variables. They are needed for testing the product effect in the step 1 model.
3. The step 1 model is estimated using the MIXED procedure. The macro parameters categorical_cov, fixed_cov and random_cov are used for the definition of the CLASS, MODEL and RANDOM statements, respectively. The output of the MIXED procedure is presented with a summary of the p-values to be considered to check the relevant assumptions. The p-values associated to the interaction between log(dose) and product (for evaluating assumption A, parallelism) and to the log(dose) effect (assumption B, dose-response) are obtained directly from the “Type 3 Tests of Fixed Effects” table in the output of the MIXED procedure. The p-values associated to the product effect (assumption C), tested at three different log(dose) levels (low, mean and high log(dose)), are derived from LSMEANS statements using the AT and PDIF options. It should be noted that the p-value provided by SAS for the product effect in the “Type 3 Tests of Fixed Effects” table corresponds to the p-value associated to the difference between products at log(dose)=0. But this does not generally make sense in our context, since this log(dose) level has no connection at all with the dose levels considered in the study.
4. The step 2 model is estimated using the MIXED procedure. The macro proceeds to the estimation of this

PhUSE 2011

model without performing any check of the assumptions based on the results from the step 1 model. They should be evaluated by the user. The output of the MIXED procedure is presented with a summary of the p-values to be considered to confirm the relevant assumptions. In this case also the p-value associated to the product effect is obtained directly from the "Type 3 Tests of Fixed Effects" table in the output of the MIXED procedure, since it does not depend any more on the log(dose) level considered. The estimates of fixed effects and their covariance matrix are generated by adding the S and COVB options, respectively, in the MODEL statement and are stored in two datasets.

- Based on the estimates of fixed effects and their covariance matrix, the relative potency and its confidence limits are calculated in a DATA step and printed. In the example below it is shown how the parameters involved in the calculations are derived from the datasets generated from the step 2 model using the MIXED procedure.

VIEWTABLE: Solution for Fixed Effects								
	Effect	product	country	Estimate	Standard Error	DF	t Value	Pr > t
1	Intercept			0.2086	0.3493	394	0.60	0.5508
2	product	R	$a_R - a_T$	-0.04660	0.09656	394	f	0.6296
3	product	T		0
4	logdose		b	0.4232	0.06985	394	6.06	<.0001
5	baseline			1.0291	0.02402	394	42.84	<.0001
6	country		1	-0.1929	0.1174	394	-1.64	0.1012
7	country		2	0.08306	0.1193	394	0.70	0.4867
8	country		3	0

VIEWTABLE: Covariance Matrix for Fixed Effects									
	Row	Effect	product	country	Col1	Col2	Col3	Col4	Col5
1	1	Intercept			0.1220	-0.00493	0	-0.02178	-0.000002
2	2	product	R		v_{11}	0.009324	v_{12}	-0.000002	0.000000
3	3	product	T		0	0	0	0	0
4	4	logdose			-0.02178	-0.000002	v_{22}	0.004879	-0.000000
5	5	baseline			-0.00227	0.000065	0	-0.00016	0.000000
6	6	country		1	-0.00788	-0.00027	0	0.000256	-0.000000
7	7	country		2	-0.00867	0.000419	0	0.000063	0.000000
8	8	country		3	0	0	0	0	0

The %RELPO macro is available in appendix with the example presented in the following section. However, the reader is invited to contact the author to check the availability of updated versions.

EXAMPLE

This example is based on simulated data from a multinational trial where the same two doses of the test (T 50 µg, T 200 µg) and of the reference (R 50 µg, R 200 µg) products were administered. 100 subjects were randomised to each of the four treatment groups. The country and the baseline value of the response variable should be accounted for in the analysis. An extract of the dataset (named "example") is shown below.


	subject	country	treatment	product	dose	baseline	y
1	1	3	T 50	T	50	3.71	6.47
2	149	2	T 200	T	200	6.96	10.01
3	229	2	R 50	R	50	5.8	6.95
4	315	3	R 200	R	200	7.37	9.24

PhUSE 2011

The variable “product” is coded using the character strings “T” for the test and “R” for the reference and the variable “dose” correspond to the original dose, as required by the %RELPOt macro. In order to calculate the relative potency and its 90% CI based on the model required, the macro parameters should be set as follows:

```
%relpot(data=example, response=y,
        fixed_cov=baseline country, random_cov=, categorical_cov=country,
        alpha=.1);
```

The results generated by the %RELPOt macro are shown below. The output of the MIXED procedure and a summary of the relevant results are presented for both steps and the estimate of relative potency is finally printed with its CI.



The summary for step 1 is reported below. Each assumption can be evaluated based on the test performed on the relevant effect included in the model. As already clarified, the difference between products is tested at three different log(dose) levels: 3.91 (=log(50)), 4.61 =(log(50)+log(200))/2 and 5.30 (=log(200)). The “Note” column includes reminders on the interpretation of the results presented. Based on the p-values reported, all the assumptions seem satisfied.

Step 1 - Summary				
Assumption	Effect	logdose	p	Note
Parallelism	logdose*product	.	0.3348	Non-significance required
Dose-response	logdose	.	<.0001	Significance required
Difference between products	product	3.91	0.3064	Non-significance required
Difference between products	product	4.61	0.6291	Non-significance required
Difference between products	product	5.30	0.7332	Non-significance required

The summary for step 2 is presented below. The assumptions on dose-response and difference between products are confirmed. Just one test is reported for assumption C, since, after excluding the interaction from the model, the product effect does not depend any more on the log(dose) level.

Step 2 - Summary			
Assumption	Effect	p	Note
Dose-response	logdose	<.0001	Significance required
Difference between products	product	0.6296	Non-significance required

The estimate of relative potency is finally printed with its 90% CI.

Relative potency		
Relative potency	Lower confidence limit	Upper confidence limit
0.89573	0.60006	1.31471

LIMITATIONS AND COMPARISONS

The approach presented assumes that no interaction terms involving the treatment (e.g., treatment * centre) are to be included in the model. Therefore it cannot be applied with models for repeated measurements, where the interaction treatment * visit is usually fitted.

Furthermore, the method by Kenward and Roger [8] for computing the denominator degrees of freedom for the tests of fixed effects (option DDFM=KR in the MODEL statement of the MIXED procedure), routinely used in the analysis of cross-over trials based on a random subject effect, is not available. Further investigation would be required in order to define the appropriate degrees of freedom f to be used in the calculation of the confidence limits for relative potency in this context.

Compared to the method presented by Finney [7], the main difference in our approach is the exclusion of the interaction term (after having checked the absence of significant deviations from parallelism in step 1) from the step 2 model, used for the estimation of relative potency. Finney estimates relative potency directly based on the results from the model including the interaction between the product and the dose. We believe that our approach is consistent with the assumption of parallel line assay, since the estimation is based on a model postulating parallel regression lines. With this regard, our two-step analysis is similar to the one suggested in the already mentioned Health Canada guidance [4], where the interaction term is excluded from the final model after having checked its non-significance.

CONCLUSION

The %RELPO macro allows to straightforwardly calculate the point estimate and the CI of relative potency based on the assumption of parallel line assay in a wide range of settings. It therefore represents a useful tool for the evaluation of therapeutic equivalence of inhaled asthma drugs according to the most recent regulatory requirements [1].

REFERENCES

1. CHMP (2009). *Guideline on the requirements for clinical documentation for Orally Inhaled Products (OIP) including the requirements for demonstration of therapeutic equivalence between two inhaled products for use in the treatment of asthma and Chronic Obstructive Pulmonary Disease (COPD) in adults and for use in the treatment of asthma in children and adolescents (CPMP/EWP/4151/00 Rev. 1)*. EMA: London.
2. Senn S (2007). *Statistical Issues in Drug Development, 2nd Edition*. Wiley: Chichester.
3. Uchizon JA, Lane JR (2007) Empirical Pharmacokinetic/Pharmacodynamic Models. In Ette EI, Williams PJ (eds.), *Pharmacometrics: the Science of Quantitative Pharmacology*. Wiley: Hoboken, NJ.
4. Bureau of Pharmaceutical Assessment (1999). *Guidance to Establish Equivalence or Relative Potency of Safety and Efficacy of a Second Entry Short-Acting Beta₂-Agonist Metered Dose Inhaler (MDI)*. Health Canada: Ottawa.
5. Govindarajulu Z (2001). *Statistical Techniques in Bioassay, 2nd Edition*. Karger: Basel.
6. García-Arieta (in press). Sensitive studies with a significant dose-response curve for inhaled corticosteroids to investigate equivalent relative potency are feasible. *British Journal of Clinical Pharmacology*.
7. Finney DJ (1978). *Statistical Method in Biological Assay, 3rd Edition*. Griffin: London.
8. Kenward MG, Roger JH (1997). Small Sample Inference for Fixed Effects from Restricted Maximum Likelihood. *Biometrics* 53: 983-997.

ACKNOWLEDGMENTS

I would like to thank Dr. Simon Day (Clinical Trials Consulting and Training Limited), Dr. Annamaria Muraro and Dr. Daniela Casula (Chiesi Farmaceutici) for suggestions and support.

CONTACT INFORMATION

Your comments and questions are valued and encouraged. Contact the author at:

Stefano Vezzoli

CROS NT S.r.l.

Via Germania 2

Verona / 37135

Work phone: +39 045 8202666

Fax: +39 045 8205875

Email: stefano.vezzoli@cros.it

Web: www.cros.it

Brand and product names are trademarks of their respective companies.

PhUSE 2011

```
%macro relpot(data=,response=,fixed_cov=,random_cov=,categorical_cov=,alpha=);

proc datasets library=work nolist;
    delete rp;;
run; quit;

data rp01;
    set &data;
    logdose=log(dose);
run;

proc means data=rp01 min max noprint;
    var logdose;
    output out=rp02 min=min max=max;
run;

data _null_;
    set rp02;
    call symput("min_logdose",put(min,best12.));
    call symput("max_logdose",put(max,best12.));
    call symput("mean_logdose",put((min+max)/2,best12.));
run;

ods output Tests3=rp03 Diffs=rp04;
proc mixed data=rp01;
    class product &categorical_cov;
    model &response = product logdose product*logdose &fixed_cov;
    %if &random_cov ne %then %do;
        random &random_cov;
    %end;
    lsmeans product / at logdose=&min_logdose pdiff;
    lsmeans product / at logdose=&mean_logdose pdiff;
    lsmeans product / at logdose=&max_logdose pdiff;
    title 'Step 1';
run;

data rp05;
    format Assumption $27. Note $25. p PVALUE6.4;
    set rp03 rp04;
    if effect='logdose*product' then assumption='Parallelism';
    else if effect='logdose' then assumption='Dose-response';
    else if effect='product' and numdf=. then assumption='Difference between
products';
    if assumption in ('Parallelism' 'Difference between products') then
note='Non-significance required';
    else if assumption='Dose-response' then note='Significance required';
    if probf ne . then p=probf;
    else p=probt;
run;

proc sort data=rp05 out=rp06 (keep=assumption note p effect logdose);
    where Assumption ne ' ';
    by descending Assumption logdose;
run;

proc print data=rp06 noobs;
    var assumption effect logdose p note;
    title 'Step 1 - Summary';
run;

ods output Tests3=rp07 SolutionF=rp08 CovB=rp09;
proc mixed data=rp01;
```

PhUSE 2011

```

class product &categorical_cov;
model &response = product logdose &fixed_cov / s covb;
%if &random_cov ne %then %do;
    random &random_cov;
%end;
title 'Step 2';
run;

data rp10;
    format Assumption $27. Note $25.;
    set rp07;
    if effect='logdose' then assumption='Dose-response';
    else if effect='product' then assumption='Difference between products';
    if assumption='Dose-response' then note='Significance required';
    else if assumption='Difference between products' then note='Non-
significance required';
run;

proc sort data=rp10 out=rp11 (keep=assumption note effect probf);
    where Assumption ne ' ';
    by descending Assumption;
run;

proc print data=rp11 noobs label;
    var assumption effect probf note;
    label probf='p';
    title 'Step 2 - Summary';
run;

data rp12;
    set rp08 rp09 end=eof;
    retain diff b t v11 v12 v22;
    if effect='product' and stderr ne . then do;
        diff=estimate;
        t=tinv(1-&alpha/2,df);
    end;
    else if effect='logdose' and stderr ne . then b=estimate;
    else if row=2 then do;
        v11=col2;
        v12=col4;
    end;
    else if row=4 then v22=col4;
    if eof then do;
        r=diff/b;
        g=(t**2)*v22/(b**2);
        lcl=(r - g*v12/v22 - (t/b) * sqrt(v11 - 2*r*v12 + (r**2)*v22 -
g*(v11-(v12**2)/v22))) / (1-g);
        ucl=(r - g*v12/v22 + (t/b) * sqrt(v11 - 2*r*v12 + (r**2)*v22 -
g*(v11-(v12**2)/v22))) / (1-g);
        exp_r=exp(r);
        exp_lcl=exp(lcl);
        exp_ucl=exp(ucl);
    end;
run;

proc print data=rp12 noobs label;
    where exp_r ne .;
    var exp_r exp_lcl exp_ucl;
    title 'Relative potency';
    label exp_r='Relative potency' exp_lcl='Lower confidence limit'
exp_ucl='Upper confidence limit';
run;

```

```

title ' ';

%mend relpot;

/* EXAMPLE */

data example (drop=i log_dose);
  format subject 3. country 1. treatment $5. product $1. dose 3.;
  do treatment='T 50', 'T 200', 'R 50', 'R 200';
    do i=1 to 100;
      subject+1;
      country=ceil(3*ranuni(43783327));
      product=scan(treatment,1);
      dose=input(scan(treatment,2),3.);
      log_dose=log(dose);
      baseline=round(5+2*rannor(0),.01);
      /* Relative potency = 0.9, Beta = 0.5 */

      y=round(baseline+log(.9)*.5*(product='R')+.5*log_dose+rannor(0),.01);
      output;
    end;
  end;
run;

%relpot(data=example,
  response=y,
  fixed_cov=baseline country,
  random_cov=,
  categorical_cov=country,
  alpha=.1);

```